Lectures on Geometry of Monge–Ampère Equations with Maple

Alexei Kushner, Valentin Lychagin and Jan Slovák

Introduction

The main goal of these lectures is to give a brief introduction to application of contact geometry to Monge–Ampère equations. These equations have the form

$$Av_{xx} + 2Bv_{xy} + Cv_{yy} + D(v_{xx}v_{yy} - v_{xy}^2) + E = 0, \tag{1}$$

were A, B, C, D and E are functions on independent variables x, y, unknown function $v = v(x, y)$, and its first derivatives v_x, v_y.

Equations of this type arise in various fields. For example, G. Monge considered such equations in connection with the problem of the optimal transportation of sand or soil. This problem was of great importance for the construction of fortifications. A modern modification of this problem has applications to mathematical economics, especially in taxations problem (Kantorovich–Monge problem [7]).

J.G. Darboux studied and applied such equations in his lectures on general theory of surfaces [3, 4, 5]. At that time geometry was a source of various types of equations. For example, the problem of reconstructing a surface with a given Gaussian curvature $K(x, y)$ is equivalent to solving the following equation:

Alexei Kushner
Lomonosov Moscow State University, GSP-2, Leninskie Gory, Moscow 119991, Russia; Moscow Pedagogical State University, 1/1 M. Pirogovskaya Str., Moscow, Russia; V. A. Trapeznikov Institute of Control Sciences of Russian Academy of Sciences, 65 Profsoyuznaya Str., Moscow 117997, Russia, e-mail: kushner@physics.msu.ru

Valentin Lychagin
V. A. Trapeznikov Institute of Control Sciences of Russian Academy of Sciences, 65 Profsoyuznaya Str., Moscow 117997, Russia; UiT Norges Arktiske Universitet, Postboks 6050, Langnes 9037, Tromso, Norway, e-mail: valentin.lychagin@uit.no

Jan Slovák
Department of Mathematics and Statistics, Masaryk University, Kotlářská 2, 611 37 Brno, Czech Republic, e-mail: slovak@muni.cz

$$v_{xx}v_{yy} - v_{xy}^2 = K(x, y)\sqrt{1 + v_x^2 + v_y^2}. \tag{2}$$

Nowadays the number of sources of Monge–Ampère equations has increased. Equations arise in physics, aerodynamics, hydrodynamics, filtration theory, in models of development of oil and gas fields, in meteorology and so on. Some of these applications will be discussed. On the other hand, as we shall see, the Monge–Ampère equations themselves generate geometric structures. For instance, some hyperbolic equations can be considered as almost product structures, and elliptic ones as almost complex structures.

The class of equations is rather wide and contains all linear and quasi-linear equations as we can see. On the other hand, it is the minimal class that contains quasilinear equations and that is closed with respect to contact transformations.

This fact was known to Sophus Lie, who applied contact geometry methods to this kind of equations. In his paper S. Lie posed some classification problems for equations with respect to contact pseudogroup. In particular, he posed the problem of equivalence of equations to the quasilinear and linear forms. This problem was solved by V.V. Lychagin and V.N. Rubtsov [19] in symplectic case and by A.G. Kushner [11] in contact case. Conditions when equations can be transformed to equations with constant coefficients by contact transformations were found by D.V. Tunitskii [21].

In 1978 V.V. Lychagin noted that the classical Monge-Ampère equations and its multi-dimensional analogues admit effective description in terms of differential forms on the space of 1-jets of smooth functions [15]. His idea was fruitful, and it generated a new approach to Monge-Ampère equations.

The lectures has the following structure.

The first lecture is an introduction to geometry of 1-jets space. We define 1-jets of scalar functions, Cartan distribution, contact transformations and contact vector fields on the 1-jets space [14, 22].

In the second lecture we describe V.V. Lychagin approach and an introduction to geometry of the Monge-Ampère equations. We follow papers [15, 16] and books [14, 17].

Third lecture is devoted to contact transformations of the Monge-Ampère equations. We consider examples of such transformations and apply them to construct multivalued solutions. We illustrate this on example of equation arising in filtration theory of two immiscible fluids (oil and water, for example) in porous media [1].

In fourth lecture we study geometrical structures associated with nondegenerated (i.e., hyperbolic and elliptic) equations. We consider also the class of so called symplectic equations and give a criterion of their linearization by symplectic transformation [18, 17].

The last, fifth lecture is devoted to tensor invariants of the Monge-Ampère equations. We construct here differential 2-forms that generalize the well known Laplace invariants. We follow the papers [11, 13].

All calculations in these lectures are illustrated in the program Maple. The Maple files can be found on the web site `d-omega.org`.

Lecture 1. Introduction to Contact Geometry

Bundle of 1-jets

Let M be an n-dimensional smooth manifold, $C^\infty(M)$ be the ring of smooth functions on M and $T^*_a M$ be the cotangent space at the point $a \in M$.

Definition 1 A 1-*jet* $[f]^1_a$ *of a function* $f \in C^\infty(M)$ *at the point* a is a pair

$$(f(a), df|_a) \in \mathbb{R} \times T^* M.$$

The set of 1-jets at the point $a \in M$ of all functions

$$J^1_a M := \left\{[f]^1_a \mid f \in C^\infty(M)\right\}$$

is a vector space with respect to operations of addition and multiplication by real numbers which are pointwise defined:

$$[f]^1_a + [g]^1_a := [f+g]^1_a, \qquad k[f]^1_a := [kf]^1_a.$$

Denote by

$$J^1 M := \mathbb{R} \times T^* M$$

the set of 1-jets of all smooth functions $f \in C^\infty(M)$ at all points $a \in M$.

This is a smooth manifold of dimension $2 \dim M + 1$ with local coordinates $x_1, \ldots, x_n\, u,\, p_1, \ldots, p_n$, where $x_1, \ldots, x_n$ are local coordinates on M, $p_1, \ldots, p_n$ are the induced coordinates on the cotangent bundle and u is the standard coordinate on $\mathbb{R}$. In other words, values of these functions at point $[f]^1_k \in J^1 M$ are the following:

$$x_i([f]^1_a) = x_i(a), \quad u([f]^1_a) = f(a), \quad p_i([f]^1_a) = f_{x_i}(a), \quad i = 1, \ldots, n. \tag{3}$$

These coordinates are called *canonical*.

In what follows we'll call $J^1 M$ the *manifold of 1-jets*, and the projection

$$\pi_1 : J^1 M \longrightarrow M, \quad \text{where} \quad \pi_1 : [f]^1_a \longmapsto a$$

the *1-jet bundle*.

Any function $f \in C^\infty(M)$ defines the following map:

$$j_1(f)\colon M \longrightarrow J^1 M, \tag{4}$$

where

$$j_1(f)\colon M \ni a \longmapsto [f]^1_a \in J^1_a M \subset J^1 M.$$

The image

$$\Gamma^1_f := j_1(f)(M) \subset J^1 M,$$

which is a smooth submanifold of $J^1 M$, is called the 1-*graph* of the function f.

Consider the following differential 1-form

$$\varkappa := du - p_1 dx_1 - \cdots - p_n dx_n$$

on the 1-jet space J^1M which we'll call *Cartan form*.

It is easy to check that this form does not depend on a choice of canonical coordinates in J^1M.

This form allows us to separate submanifolds of the form $\Gamma^1_f \subset J^1M$ from arbitrary submanifolds of dimension n by observation that

$$\varkappa|_{\Gamma^1_f} = 0,$$

for any $f \in C^\infty(M)$. Indeed,

$$\varkappa|_{\Gamma^1_f} = df - f_{x_1} dx_1 - \cdots - f_{x_i} dx_i = 0.$$

On the other hand, if a submanifold $N \subset J^1M$ is a graph of section $s \colon M \longrightarrow J^1M$, i.e. $\pi_1 \colon N \longrightarrow M$ is a diffeomorphism, and

$$\varkappa|_N = 0,$$

then one can easily check that $N = \Gamma^1_f$ for some smooth function $f \in C^\infty(M)$.

This observation shows that zeroes of the Cartan form (but not the form itself) are important to distinguish 1-graphs from arbitrary submanifolds in J^1M.

Denote by C the $2n$-dimensional distribution (Cartan distribution) on J^1M given by zeroes of the Cartan form:

$$C : J^1M \ni \theta \longmapsto C(\theta) := \ker \varkappa_\theta \subset T_\theta(J^1M).$$

In the dual way, the Cartan distribution can be defined by vector fields tangent to this distribution. Namely, vector fields

$$\partial_{x_1} + p_1\partial_u, \ldots, \partial_{x_n} + p_n\partial_u, \partial_{p_1}, \ldots, \partial_{p_n}$$

give us a local basis in the module of vector fields tangent to C. This module will be denoted by $D(C)$.

Then a submanifold $N \subset J^1M$ is a graph of a smooth function if and only if

1. N is an integral submanifold of the Cartan distribution, and
2. $\pi_1 \colon N \to M$ is a diffeomorphism.

Remind that a contact structure on an odd dimensional manifold K, $\dim K = 2k + 1$, consists of $2k$-dimensional distribution P on K such that

$$\lambda \wedge (d\lambda)^k \neq 0$$

for any differential 1-form λ, such that locally $P = \ker\, \lambda$.

In our case, we have

$$\varkappa \wedge (d\varkappa)^n \neq 0$$

and therefore the Cartan distribution defines the contact structure on the manifold of 1-jets J^1M.

Contact transformations

A transformation Φ of the space J^1M is called *contact*, if it preserves the Cartan distribution, i.e.

$$\Phi_*(C) = C.$$

In terms of the Cartan form, a transformation Φ is contact if

$$\Phi^*(\varkappa) = h_\Phi \varkappa \tag{5}$$

for some function h_Φ, or equivalently

$$\Phi^*(\varkappa) \wedge \varkappa = 0.$$

Examples of contact transformations

1. Translations:

$$(x_1, x_2, u, p_1, p_2) \longmapsto (x_1 + \alpha_1, x_2 + \alpha_2, u + \beta, p_1, p_2),$$

where α_1, α_2 and β are constants.

2. The Legendre transformation:

$$(x_1, x_2, u, p_1, p_2) \longmapsto (p_1, p_2, u - x_1 p_1 - x_2 p_2, -x_1, -x_2).$$

3. Partial Legendre's transformation:

$$(x_1, x_2, u, p_1, p_2) \longmapsto (p_1, x_2, u - p_1 x_1, -x_1, p_2).$$

Infinitesimal versions of contact transformations are contact vector fields.

A vector field X on J^1M is called *contact* if its local translation group consists of contact transformations.

It means that

$$\Phi_t^*(\varkappa) = \lambda_t \varkappa \tag{6}$$

for some function λ_t on J^1M. Here Φ_t are shifts along vector field X.

After differentiating both parts of (6) by t at $t = 0$, we get:

$$\frac{d}{dt}\bigg|_{t=0} \left(\Phi_t^*(\varkappa)\right) = \left(\frac{d\lambda}{dt}\bigg|_{t=0}\right)\varkappa.$$

The left hand side of the equation is the Lie derivative $L_X(\varkappa)$ of the Cartan form in the direction of the vector field X and, therefore, we get

$$L_X(\varkappa) = h\varkappa,$$

where h is a function on J^1M.

Multiplying both parts of the last equation by $\varkappa$ we get:

$$L_X(\varkappa) \wedge \varkappa = 0. \tag{7}$$

In canonical coordinates, each contact vector field has the form

$$X_f = -\sum_{i=1}^{n} \frac{\partial f}{\partial p_i}\frac{\partial}{\partial x_i} + \left(f - \sum_{i=1}^{n} p_i \frac{\partial f}{\partial p_i}\right)\frac{\partial}{\partial u} + \sum_{i=1}^{n}\left(\frac{\partial f}{\partial x_i} + p_i \frac{\partial f}{\partial u}\right)\frac{\partial}{\partial p_i}$$

for some function f which is called *generating function* of the contact vector field. Note that

$$\varkappa(X_f) = f.$$

Maple Code: main operation on $J^1\mathbb{R}^2$

1. Load libraries:

```
with(DifferentialGeometry): with(JetCalculus):
```

2. Set jet notation, declare coordinates on the manifold M and generate coordinates on the 1-jet space:

```
Preferences("JetNotation", "JetNotation2"):
DGsetup( [x1,x2],[u], M, 1, verbose);
```

3. Generate the Cartan form:

```
kappa:= convert(Cu[0,0],DGform);
```

4. Define partial Legendre transformation:

```
PartLegendre:=Transformation(M,M,[x1=-u[1,0],x2=x2,
u[0,0]=u[0,0]-u[1,0]*x1, u[1,0]=x1, u[0,1]=u[0,1]]);
```

5. Apply this transformation to the Cartan form:

```
Pullback(PartLegendre,kappa);
```

6. Prolongation of transformations from J^0M to J^1M:

```
Phi:=Transformation(M,M,
[x1=x2,x2=x1+x2,u[0,0]=-u[0,0]]);
Prolong(Phi,1);
```

7. Define the contact vector field X_f with generating function $f = p_2$:

```
X:=GeneratingFunctionToContactVector(u[0, 1]);
```

8. Prolongation of vector fields from the plane $M = \mathbb{R}^2$ to J^1M:

```
Y:=evalDG(-x2*D_x1+x_1*D_x2);
Prolong(Y,1);
```

Lecture 2. Geometrical Approach to Monge–Ampère Equations

Nonlinear Second Order Differential Operators

Following [15], any differential n-form ω on J^1M is associated with the differential operator

$$\Delta_\omega : C^\infty(M) \longrightarrow \Omega^n(M),$$

which acts in the following way:

$$\Delta_\omega(v) := j_1(v)^*(\omega), \tag{8}$$

where (see formula (4))

$$j_1(v)^* : \Omega^n(J^1M) \longrightarrow \Omega^n(M).$$

This construction does not cover all nonlinear second order differential operators, but only a certain subclass of them.

Examples

1. The differential 1-form on $J^1\mathbb{R}$

$$\omega = (1 - x^2)dp + (\lambda u - xp)\,dx,$$

where

$$\lambda = \frac{a^2}{b^2},$$

generates the Lissajou differential operator

$$\Delta_\omega(y) = \left((1 - x^2)y'' - xy' + \frac{a^2}{b^2}y\right)dx. \tag{9}$$

Indeed,

$$\Delta_\omega(v) = (1-x^2)d\left(y'\right) + \left(-xy' + \frac{a^2}{b^2}y\right)dx$$
$$= \left((1-x^2)y'' - xy' + \frac{a^2}{b^2}y\right)dx.$$

2. The differential 2-form on $J^1\mathbb{R}^2$

$$\omega = dp_1 \wedge dp_2$$

generates the Hesse operator

$$\Delta_\omega(v) = (\det \operatorname{Hess} v)\, dx_1 \wedge dx_2. \tag{10}$$

Indeed,

$$\begin{aligned}\Delta_\omega(v) &= d\left(v_{x_1}\right) \wedge d\left(v_{x_2}\right)\\ &= \left(v_{x_1x_1}dx_1 + v_{x_1x_2}dx_2\right) \wedge \left(v_{x_2x_1}dx_1 + v_{x_2x_2}dx_2\right)\\ &= \left(v_{x_1x_1}v_{x_2x_2} - v_{x_1x_2}^2\right)dx_1 \wedge dx_2\\ &= (\det \operatorname{Hess} v)\, dx_1 \wedge dx_2,\end{aligned}$$

where $\operatorname{Hess} v$ is the Hessian of the function v.

3. The differential 3-form

$$\omega = p_1 dp_1 \wedge dx_2 \wedge dx_3 - dx_1 \wedge dp_2 \wedge dx_3 - dx_1 \wedge dx_2 \wedge dp_3 \tag{11}$$

on $J^1\mathbb{R}^3$ produces the von Karman differential operator

$$\left(v_x v_{xx} - v_{yy} - v_{zz}\right) dx \wedge dy \wedge dz,$$

where $x = x_1,\ y = x_2,\ z = x_3$.

4. The differential 2-form

$$\omega = dp_1 \wedge dx_2 - dp_2 \wedge dx_1$$

on $J^1\mathbb{R}^2$ represents the 2-dimensional Laplace operator

$$\Delta_\omega(v) = \left(v_{xx} + v_{yy}\right) dx \wedge dy,$$

where $x = x_1,\ y = x_2$.

5. Two differential 2-forms

$$\omega = dx_1 \wedge du \qquad \text{and} \qquad \varpi = p_2 dx_1 \wedge dx_2 \tag{12}$$

on $J^1\mathbb{R}^2$ generate the same operator:

$$\Delta_\omega(v) = dx_1 \wedge \left(v_{x_1} dx_1 + v_{x_2} dx_2\right) = v_{x_2}\, dx_1 \wedge dx_2,$$
$$\Delta_\varpi(v) = v_{x_2}\, dx_1 \wedge dx_2.$$

6. Any differential n-form
$$\omega = \varkappa \wedge \alpha + d\varkappa \wedge \beta \tag{13}$$
on J^1M, where $\alpha \in \Omega^{n-1}\left(J^1M\right)$, $\beta \in \Omega^{n-2}\left(J^1M\right)$ and $\varkappa$ is the Cartan form, gives the zero operator.

All differential operators Δ_ω generate differential equations of second order:
$$\Delta_\omega(v) = 0. \tag{14}$$

For example, operator (9) generates Lissajou equation
$$(1 - x^2)y'' - xy' + \frac{a^2}{b^2}y = 0. \tag{15}$$

Note that the differential operators Δ_ω and $\Delta_{h\omega}$ generate the same equation for each non-zero function h.

The equations (14) are called *Monge–Ampère equations* [15].

The following observation justifies this definition: being written in local canonical contact coordinates on J^1M, the operators Δ_ω have the same type of nonlinearity as the Monge-Ampère equations.

Namely, the nonlinearity involves the determinant of the Hesse matrix and its minors. For instance, in the case $n = 2$, for
$$\begin{aligned}\omega =& E dx_1 \wedge dx_2 + B\left(dx_1 \wedge dp_1 - dx_2 \wedge dp_2\right) + \\ & C dx_1 \wedge dp_2 - A dx_2 \wedge dp_1 + D dp_1 \wedge dp_2.\end{aligned} \tag{16}$$

we get classical Monge-Ampère equations
$$A v_{xx} + 2B v_{xy} + C v_{yy} + D(v_{xx}v_{yy} - v_{xy}^2) + E = 0. \tag{17}$$

An advantage of this approach is the reduction of the order of the jet space: we use the simpler space J^1M instead of the space J^2M where Monge–Ampère equations should be *ad hoc* as second-order partial differential equations [22].

The differential equation which is associated with a differential n-form ω will be denote by $\mathcal{E}_\omega$:
$$\mathcal{E}_\omega := \{\Delta_\omega(v) = 0\}.$$

The following Maple code generates the corresponding differential operator Δ_ω for a differential 2-form ω on $J^1\mathbb{R}^2$.

Maple Code: $\omega \longmapsto \Delta_\omega$

```
with(DifferentialGeometry): with(JetCalculus):
Preferences("JetNotation", "JetNotation2"):
DGsetup( [x1,x2],[u], M, 1);
DGsetup( [x,y], N, verbose);
```

Construct the differential operator Δ:

```
Delta := proc(z, h)
    Pullback(Prolong(Transformation(N,M,
    [x1=x,x2=y,u[0,0]=h]),2),z);
end proc;
```

Define a differential 2-form:

```
omega:=evalDG(dx1 &w du[1,0]-dx2 &w du[0,1]);
```

Apply the differential operator to this differential form $\omega = dx_1 \wedge dp_1 - dx_2 \wedge dp_2$:

```
simplify(Delta(omega,v(x,y)),size);
```

As a result we get the differential operator

$$2\frac{\partial^2}{\partial y \partial x} dx \wedge dy.$$

Multivalued solutions of Monge–Ampère equations

Let v be a classical solution of the Monge–Ampère equation $\mathcal{E}_\omega$, i.e. $\Delta_\omega(v) = 0$. Then

$$j_1(v)^*(\omega) = 0.$$

It means that the restriction of the differential form ω to 1-graph of the function v is zero:

$$\omega \mid_{\Gamma_v^1} = 0.$$

An n-dimensional submanifold $L \subset J^1M$ is called a *multivalued solution* of Monge–Ampère equation if

1. L is an integral manifold of the Cartan distribution, i.e. the restriction of the Cartan form to L is zero: $\varkappa \mid_L = 0$;
2. the restriction of the differential n-form ω to L is zero, too: $\omega \mid_L = 0$.



Fig. 1 Multivalued solutions of the Lissajou equation for $a = 3, b = 2$.



Fig. 2 Multivalued solutions of the Lissajou equation for $a = 1, b = \sqrt{2}$ is a curve, everywhere dense in the square (Lissajou's Black Square)

Examples: Multivalued solutions

1. Parameterized curves

$$L = \left\{ x = \sin\, bt,\ y = \cos\, at,\ p = -\frac{a \sin\, at}{b \cos\, bt} \right\}$$

in the space $J^1\mathbb{R}$ are multivalued solutions of the Lissajou equation

$$(1 - x^2)y'' - xy' + \frac{a^2}{b^2}y = 0. \tag{18}$$

Indeed, the restriction of the differential 1-form

$$\omega = (1 - x^2)dp + \left(\frac{a^2}{b^2} y - xp \right) dx$$

on the curve L is zero. The projections of these curves on the plane $(x,\ y)$ are well known Lissajou curves (see Fig. 1, 2).

2. Projections of multivalued solutions of the Monge equation

$$v_{xx}v_{yy} - v_{xy}^2 = \sqrt{1 + v_x^2 + v_y^2}$$

to the space $\mathbb{R}^3$ with coordinates x, y, v are spheres with radius 1 (see equation (2).

3. Projections of multivalued solutions of the equation

$$v_{xx}v_{yy} - v_{xy}^2 = 0 \tag{19}$$

to the space $\mathbb{R}^3$ with coordinates x, y, v are deployable surfaces.

Effective forms

Last two examples (12) and (13) show that the constructed map

"differential n-forms" $\rightarrow$ "differential operators"

has a huge kernel.

This kernel consists of differential forms that vanish on any integral manifold of the Cartan distribution. All such forms have form (13) (see [14]).

Let's find a submodule of the module $\Omega^2(J^1M)$ of differential 2-forms such that the map is bijective ($\dim M = 2$).

Differential 2-form $\omega \in \Omega^2(J^1M)$ is called *effective* if

1. $X_1 \rfloor \omega = 0$;
2. $\omega \wedge d\varkappa = 0$.

Here X_1 is the contact vector field with generating function 1. In canonical coordinates (3)

$$X_1 = \partial_u.$$

The first condition means that coordinate representation of ω does not contain terms $du \wedge *$, and, therefore $\omega \neq \varkappa \wedge \alpha$ for some differential 1-form α. Second condition means that $\omega \neq \beta d\varkappa$, for a function β.

The module of effective differential 2-forms will be denoted by $\Omega^2_\epsilon(J^1M)$.

There is the projection p which maps module $\Omega^2(J^1M)$ to the module $\Omega^2(C)$ of "differential forms" on the Cartan distribution.

Namely, define

$$p : \Omega^2(J^1M) \longrightarrow \Omega^2(C)$$

as follows:

$$p(\omega) := \omega - \varkappa \wedge (X_1 \rfloor \omega).$$

Here $\Omega^2(J^1M)$ and $\Omega^2(C)$ are modules of 2-forms on the 1-jet manifold J^1M and on the Cartan distribution C respectively. Remark that

$$X_1 \rfloor p(\omega) = 0,$$

i.e. 2-form $p(\omega) \in \Omega^2(C)$.

Theorem 1 *Any differential 2-form $\omega \in \Omega^2(C)$ has the unique representation*

$$\omega = \omega_\epsilon + \beta d\varkappa, \tag{20}$$

where $\omega_\epsilon \in \Omega^2_\epsilon(J^1M)$ is an effective 2-form and β is a function.

Proof In our case the Cartan distribution C is 4-dimensional. The exterior differential of the Cartan form is non-degenerated 2-form on each Cartan subspace, i.e. $d\varkappa_\theta$ is a symplectic structure on $C(\theta)$ for any $\theta \in J^1M$. Therefore, formula

$$\omega \wedge d\varkappa = \beta d\varkappa \wedge d\varkappa$$

uniquely defines a function β. Define now differential form

$$\omega_\epsilon = \omega - \beta d\varkappa.$$

Since $\omega_\epsilon \wedge d\varkappa = 0$, the form ω_ϵ is effective. □

The constructed differential form ω_ϵ is called the *effective part* of the differential form ω.

Define the operator

$$\mathrm{Eff} : \Omega^2(J^1M) \longrightarrow \Omega^2_\epsilon(J^1M), \quad \mathrm{Eff}(\omega) := (p(\omega))_\epsilon,$$

which for any differential 2-form ω on the space J^1M gives its effective part.

It is obvious that differential 2-forms ω and $\mathrm{Eff}(\omega)$ generate the same Monge–Ampère equations.

In canonical coordinates

$$d\varkappa = dx_1 \wedge dp_1 + dx_2 \wedge dp_2$$

and any effective differential 2-form has the following representation:

$$\begin{aligned} \omega =& E dx_1 \wedge dx_2 + B\,(dx_1 \wedge dp_1 - dx_2 \wedge dp_2) + \\ & C dx_1 \wedge dp_2 - A dx_2 \wedge dp_1 + D dp_1 \wedge dp_2, \end{aligned} \tag{21}$$

where A, B, C, D and E are smooth functions on J^1M. This form corresponds to equation (17).

The following Maple code contains two procedures which generate effective parts of differential 2-forms.

Maple Code: $\omega \longmapsto \omega_\epsilon$

1. Projection of a 2-form to the Cartan distribution:

```
ProjC:=proc (omega)
    GeneratingFunctionToContactVector(1);
    evalDG(omega-kappa &w Hook(evalDG(D_u[0,0]),omega));
end proc:
```

2. Calculation of effective parts of a differential 2-forms:

```
Eff:=proc (omega)
    evalDG(evalDG(omega-kappa &w Hook(evalDG(D_u[0,0]),
    omega))-(solve(op(Tools:-DGinfo(evalDG(g*Omega&w Omega-
    (evalDG(omega-kappa &w Hook(evalDG(D_u[0,0]),omega)))
    &w Omega),"CoefficientSet")),g))*Omega);
end proc:
```

Lecture 3. Contact transformations of Monge–Ampère equations

By the definition, contact transformations preserve the Cartan distribution and multiply the Cartan form $\varkappa$ by a function (see formula (5)).

Therefore contact transformations do not preserve the contact vector field X_1 in general. Because of this the image of an effective differential form can be not effective.

Let $\Phi : J^1M \to J^1M$ be a contact transformation and ω be an effective differential 2-form. Then by image of differential 2-form ω we shall understand the effective differential form $\mathrm{Eff}(\Phi^*(\omega))$.

Two Monge–Ampère equations $\mathcal{E}_\omega$ and $\mathcal{E}_\varpi$ are *contact equivalent* if there exist a contact transformation Φ such that $\varpi = h\mathrm{Eff}(\Phi^*(\omega))$ for some function h.

Theorem 2 *If two equations $\mathcal{E}_\omega$ and $\mathcal{E}_\varpi$ are contact equivalent, then their contact transformation maps multivalued solutions of one to multivalued solutions of the other.*

Note that in general contact transformations do not preserve the class of classical solutions: classical solutions can transform to multivalued solutions and vice versa.

Examples of linearization of equations by contact transformations

1. The Von Karman equation
$$v_{x_1} v_{x_1 x_1} - v_{x_2 x_2} = 0 \tag{22}$$
becomes the linear equation
$$x_1 v_{x_2 x_2} + v_{x_1 x_1} = 0 \tag{23}$$
after Legendre transformation (24).
The last equation is known as the *Triccomi* equation.
2. Equation
$$\det \operatorname{Hess} v = 1$$
is generated by the effective differential 2-form
$$\omega = dp_1 \wedge dp_2 - dx_1 \wedge dx_2 .$$
After the partial Legendre transformation
$$\Phi : (x_1,\ x_2,\ u,\ p_1,\ p_2) \mapsto (p_1,\ x_2,\ u - p_1 x_1,\ -x_1,\ p_2)$$
this form becomes
$$\omega = dx_2 \wedge dp_1 - dx_1 \wedge dp_2,$$
and corresponds to the Laplace equation
$$v_{x_1 x_1} + v_{x_2 x_2} = 0.$$
3. Quasi-linear equation:
$$A\left(v_x, v_y\right) v_{xx} + 2B\left(v_x, v_y\right) v_{xy} + C\left(v_x, v_y\right) v_{yy} = 0.$$
This equation is represented by the following effective form
$$\omega = B\left(p_1, p_2\right)\left(dx_1 \wedge dp_1 - dx_2 \wedge dp_2\right) + C\left(p_1, p_2\right) dx_1 \wedge dp_2 - A\left(p_1, p_2\right) dx_2 \wedge dp_1 .$$
After the Legendre transformation
$$\Phi : (x_1,\ x_2,\ u,\ p_1,\ p_2) \mapsto (p_1,\ p_2,\ u - p_1 x_1 - p_2 x_2,\ -x_1,\ -x_2,) \tag{24}$$
we get the following effective form
$$\begin{aligned}\varphi^*(\omega) = {} & B\left(-x_1, -x_2\right)\left(dx_1 \wedge dp_1 - dx_2 \wedge dp_2\right) + \\ & - A\left(-x_1, -x_2\right) dx_1 \wedge dp_2 + C\left(-x_1, -x_2\right) dx_2 \wedge dp_1,\end{aligned}$$
which corresponds to the linear equation:

$$-A\,(-x_1,-x_2)\,v_{x_2x_2} + 2B\,(-x_1,-x_2)\,v_{x_1x_2} - \;C\,(-x_1,-x_2)\,v_{x_1x_1} = 0.$$

Example

The following equation arises in filtration theory of two immiscible fluids in porous media [1]:

$$u_{xy} - u_x u_{yy} = 0. \tag{25}$$

It is used for finding a strategy to control wave fronts in the development of oil fields.

The corresponding differential 2-form is

$$\omega = 2p_1 dp_2 \wedge dx_1 + dx_1 \wedge dp_1 - dx_2 \wedge dp_2,$$

where $x_1 = x$, $x_2 = y$. Applying the Legendre transformation

$$\Phi : (x_1, x_2, u, p_1, p_2) \longmapsto (p_1, p_2, u - x_1 p_1 - x_2 p_2, -x_1, -x_2)$$

we get the following differential 2-form

$$\Phi^*(\omega) = 2x_1 dx_2 \wedge dp_1 + dx_1 \wedge dp_1 - dx_2 \wedge dp_2.$$

This form corresponds to the linear equation

$$u_{x_1x_2} - x_1 u_{x_1x_1} = 0. \tag{26}$$

The general solution of the last equation is

$$u(x_1, x_2) = e^{-x_2} F_1(x_1 e^{x2}) + F_2(x_2), \tag{27}$$

where F_1 and F_2 are arbitrary functions. Differentiating both sides of (27), we get:

$$\begin{aligned} u_{x_1} &= F_1'(x_1 e^{x2}),\\ u_{x_2} &= -e^{-x_2} F_1(x_1 e^{x2}) - F_1'(x_1 e^{x2}) x_1 + F_2'(x_2). \end{aligned}$$

Thus, solution (27) generate a surface $L \subset J^1 M$:

$$L : \begin{cases} u - e^{-x_2} F_1(x_1 e^{x2}) + F_2(x_2) = 0,\\ p_1 - F_1'(x_1 e^{x2}) = 0,\\ p_2 + e^{-x_2} F_1(x_1 e^{x2}) + F_1'(x_1 e^{x2}) x_1 - F_2'(x_2) = 0. \end{cases}$$

Applying the inverse Legendre transformation

$$\Phi^{-1} : (x_1, x_2, u, p_1, p_2) \longmapsto (-p_1, -p_2, u - x_1 p_1 - x_2 p_2, x_1, x_2)$$

to L, we get multivalued solutions of equation (25) in parametric form:

$$\Phi^{-1}(L):\begin{cases} u - x_1 p_1 - x_2 p_2 - e^{p_2} F_1(-p_1 e^{-p_2}) + F_2(-p_2) = 0, \\ x_1 - F_1'(-p_1 e^{-p_2}) = 0, \\ x_2 + e^{p_2} F_1(-p_1 e^{-p_2}) + p_1 F_1'(-p_1 e^{-p_2}) + F_2'(-p_2) = 0. \end{cases} \tag{28}$$



Fig. 3 Projection of the multivalued solution $\mathcal{L}$ to the space x, y, u for $k(a) = a^9 - 20a^5$ and $r(b) = b^{0.01}$

In order to simplify the last formula, we introduce new parameters

$$a = -p_1 e^{-p_2}, \qquad b = -p_2,$$

and new functions

$$k(a) = F_1(a), \qquad r(b) = F_2(b).$$

In these notation, multivalued solutions of equation (25) takes the form:

$$\mathcal{L}:\begin{cases} x = k'(a), \\ y = e^{-b}(ak'(a) - k(a)) - r'(b), \\ u = (b+1)e^{-b}(k(a) - ak'(a)) + br'(b) - r(b), \\ p_1 = -ae^{-b}, \\ p_2 = -b, \end{cases}$$

where $k(a)$ and $r(b)$ are arbitrary functions.

Maple Code: Equation $u_{xy} - u_x u_{yy} = 0$

Define coordinates on M:

```
with(DifferentialGeometry): with(JetCalculus):
Preferences("JetNotation", "JetNotation2"):
DGsetup( [x1,x2],[u], M, 1);
DGsetup( [x,y],N,1);
```

Construct the differential operator Δ:

```
Delta := proc(z, h)
    description "M-A operator";
    Pullback(Prolong(Transformation
    (N,M,[x1=x,x2=y,u[0,0]=h]),2),z);
end proc;
```

Define the differential 2-form ω:

```
omega:=evalDG(2*u[1,0]*du[0,1] &w dx1 +
dx1 &w du[1,0]-dx2 &w du[0,1]);
```

$$\omega = 2p_1 dp_2 \wedge dx_1 + dx_1 \wedge dp_1 - dx_2 \wedge dp_2,$$

The Legendre transformation:

```
Legendre:=Transformation(M,M,[x1=u[1,0],x2=u[0,1],
u[0,0]=u[0,0]-x1*u[1,0]-u[0,1]*x2, u[1,0]=-x1, u[0,1]=-x2]):
```

Apply the Legendre transformation to ω:

```
omega1:=Pullback(Legendre,omega);
```

Construct the differential operator Δ_{ω_1}:

```
Delta(omega1,u(x,y));
```

$$\left(2\left(\frac{\partial^2}{\partial y \partial x} u(x,y)\right) - 2x\left(\frac{\partial^2}{\partial x^2} u(x,y)\right)\right) dx \wedge dy$$

Check solution:

```
sub:={u(x,y)=exp(-y)*F1(x*exp(y))+F2(y)};
eval(diff(u(x, y), x, y)-x*diff(u(x, y), x, x), sub);
```

$$0$$

Inverse Legendre transformation:

```
InvLegendre:=InverseTransformation(Legendre):
```

Apply this transformation to the surface L:

```
z1:=convert(u(x1,x2)-exp(-x2)*F1(x1*exp(x2))+F2(x2),DGjet):

z2:=convert(diff(u(x1,x2)-exp(-x2)*
F1(x1*exp(x2))+F2(x2),x1),DGjet):

z3:=convert(diff(u(x1,x2)-exp(-x2)*
F1(x1*exp(x2))+F2(x2),x2),DGjet):

u1:=Pullback(InvLegendre,z1):

u2:=Pullback(InvLegendre,z2):

u3:=Pullback(InvLegendre,z3):
```

As a result we get formula (28).
Check that $\mathcal{L}$ is a multivalued solution of equation (25), i.e. $\omega \mid \mathcal{L} = 0$:

```
DGsetup( [x1,x2,u,p1,p2], M);
DGsetup( [a,b], N);

omega:=evalDG(2*p1*dp2 &w dx1 + dx1 &w dp1-dx2 &w dp2):

NtoM:=Transformation(N,M,[x1=diff(k(a),a),
x2=exp(-b)*(a*diff(k(a),a)-k(a))-diff(r(b),b),
u=(b+1)*exp(-b)*(-a*diff(k(a),a)+k(a))+b*diff(r(b),b)-r(b),
p1=-a*exp(-b), p2=-b]):

Pullback(NtoM,omega);
```

$$0$$

Visualisation of the multivalued solution $\mathcal{L}$:

```
plot3d(eval([diff(k(a),a),exp(-b)*(a*diff(k(a),a)-k(a))
-diff(r(b),b), (b+1)*exp(-b)*(-a*diff(k(a),a)+k(a))+
b*diff(r(b),b)-r(b)], {k(a)=a^9-20*a^5,r(b)=b^0.01}),
a = -1 .. 1, b = -6 .. 6);
```

Lecture 4. Geometrical Structures

Pfaffians

First of all we remark that the restriction of the differential 2-form $d\varkappa$ on the Cartan distribution

$$\Omega = d\varkappa\,|_{C}$$

defines a symplectic structure on Cartan space $C(\theta) \subset T_\theta(J^1)M$, $\theta \in J^1M$.

Using this structure and an effective 2-form $\omega \in \Omega^2_\epsilon(J^1M)$ we define function $\mathrm{Pf}(\omega)$, called *Pfaffian*, in the following way [19]:

$$\mathrm{Pf}(\omega)\,\Omega \wedge \Omega = \omega \wedge \omega. \tag{29}$$

This is a correct construction because $\omega \wedge \omega$ and $\Omega \wedge \Omega$ are 4-forms on the 4-dimensional Cartan distribution.

In the case when

$$\begin{aligned} \omega =& E dx_1 \wedge dx_2 + B\,(dx_1 \wedge dp_1 - dx_2 \wedge dp_2) + \\ & C dx_1 \wedge dp_2 - A dx_2 \wedge dp_1 + D dp_1 \wedge dp_2, \end{aligned} \tag{30}$$

we get

$$\mathrm{Pf}(\omega) = B^2 + DE - AC.$$

We say that the Monge-Ampère equation $\mathcal{E}_\omega$ is *hyperbolic, elliptic* or *parabolic* at a domain $\mathcal{D} \subset J^1M$ if the function $\mathrm{Pf}(\omega)$ is negative, positive or zero at each point of $\mathcal{D}$, respectively.

If the Pfaffian changes the sign in some points of $\mathcal{D}$, then the equation $\mathcal{E}_\omega$ is called a *mixed type* equation (see [9]).

The hyperbolic and elliptic equations are called *nondegenerate*.

Maple Code: Pfaffian

```
kappa:=convert(Cu[0,0],DGform):
Omega:=ExteriorDerivative(kappa):
omega:=evalDG(dq1 &w du[1,0]+ du[0,0] &w du[0,1]):

Pf:=proc (omega)
solve(op(DGinfo(evalDG(z*Omega &w Omega-omega &w omega),
"CoefficientSet")),z)
end proc:
```

For example, the Pfaffian of the differential 2-form

$$\omega = dx_1 \wedge dp_1 - dx_2 \wedge dp_2$$

which corresponds to wave equation $u_{xy} = 0$ is equal to -1, and as we know this equation is hyperbolic.

The Pfaffian of the differential 2-form

$$\omega = dx_1 \wedge dp_2 - dx_2 \wedge dp_1$$

which corresponds to Laplace equation $u_{xx} + u_{yy} = 0$ is equal to 1. Indeed,

```
omega:=evalDG(dx1 &w du[1,0]-dx2 &w du[0,1]):
Pf(omega);
```

$$-1$$

```
omega:=evalDG(dx1 &w du[0,1]-dx2 &w du[1,0]):
Pf(omega);
```

$$1$$

Fields of endomorphisms

The standard linear algebra allows us to construct a field of endomorphisms

$$A_\omega \colon D(C) \longrightarrow D(C)$$

which is associated with an effective 2-form ω. Here $D(C)$ is the module of vector fields tangent to C.

Namely, the 2-form Ω is non-degenerated on C and the operator A_ω is uniquely determined by the following formula [18]:

$$A_\omega X \rfloor\ \Omega = X \rfloor \omega \tag{31}$$

for all vector fields X tangent to C.

Proposition 1 *Operators A_ω satisfy the following properties:*

1. $\Omega(A_\omega X, X) = 0$.
2. $\Omega(A_\omega X, Y) = \Omega(X, A_\omega Y)$.

Proof 1. $\Omega(A_\omega X, X) = \omega(X, X) = 0$.
2. $\Omega(A_\omega X, Y) = \omega(X, Y) = -\omega(Y, X) = -\Omega(A_\omega Y, X) = \Omega(X, A_\omega Y)$. □

Proposition 2 *The squares of operators A_ω are scalar and*

$$A_\omega^2 + \mathrm{Pf}(\omega) = 0. \tag{32}$$

Proof First of all

$$A_\omega X\rfloor(\omega\wedge\Omega) = (A_\omega X\rfloor\omega)\wedge\Omega + \omega\wedge(A_\omega X\rfloor\Omega).$$

Using Proposition 1,

$$\begin{aligned} X\rfloor(A_\omega X\rfloor(\omega\wedge\Omega)) =& \omega(A_\omega X, X)\Omega - (A_\omega X\rfloor\omega)\wedge(X\rfloor\Omega)\\ &+ (X\rfloor\omega)\wedge(A_\omega X\rfloor\Omega) + \Omega(A_\omega X, X)\omega\\ &= \Omega(A_\omega^2 X, X)\Omega - (A_\omega^2 X\rfloor\Omega)\wedge(X\rfloor\Omega)\\ &+ (A_\omega X\rfloor\Omega)\wedge(A_\omega X\rfloor\Omega) + \Omega(A_\omega X, X)\omega\\ &= -(A_\omega^2 X\rfloor\Omega)\wedge(X\rfloor\Omega). \end{aligned}$$

Since ω is effective, $\omega\wedge\Omega = 0$. Then

$$(A_\omega^2 X\rfloor\Omega)\wedge(X\rfloor\Omega) = 0,$$

i.e. differential 1-forms $A_\omega^2 X\rfloor\Omega$ and $X\rfloor\Omega$ are linearly dependent. Therefore the square of the operator A_ω is a scalar: $A_\omega^2 = \alpha$.

Let $X \in D(C)$ be an arbitrary vector field. Applying the operators $A_\omega X\rfloor$ and $X\rfloor$ to both parts of formula (29) we get

$$\mathrm{Pf}(\omega)(A_\omega X\rfloor\Omega)\wedge(X\rfloor\Omega) = (A_\omega X\rfloor\omega)\wedge(X\rfloor\omega) = (\alpha X\rfloor\Omega)\wedge(A_\omega X\rfloor\Omega).$$

Then

$$(\mathrm{Pf}(\omega)+\alpha)(A_\omega X\rfloor\Omega)\wedge(X\rfloor\Omega) = 0. \tag{33}$$

Suppose that $(A_\omega X\rfloor\Omega)\wedge(X\rfloor\Omega) = 0$. Then the vector fields X and $A_\omega X$ are linearly dependent. Since X is an arbitrary vector field we see that the operator A_ω is scalar, i.e. $A_\omega X = \lambda X$ for any X. Then

$$X\rfloor\omega = A_\omega X\rfloor\Omega = \lambda X\rfloor\Omega.$$

Therefore $\omega = \lambda\Omega$, which is impossible. So, from (33) it follows that $\mathrm{Pf}(\omega)+\alpha = 0$, i.e. $A_\omega^2 + \mathrm{Pf}(\omega) = 0$. □

Let's find a coordinate representation of the operator A_ω. Let

$$\frac{\partial}{\partial x_1} + p_1\frac{\partial}{\partial u}, \quad \frac{\partial}{\partial x_2} + p_2\frac{\partial}{\partial u}, \quad \frac{\partial}{\partial p_1}, \quad \frac{\partial}{\partial p_2} \tag{34}$$

be a local basis of the module $D(C)$. Then formula (31) gives:

$$A_\omega = \begin{Vmatrix} B & -A & 0 & -D\\ C & -B & D & 0\\ 0 & E & B & C\\ -E & 0 & -A & -B \end{Vmatrix} \tag{35}$$

in this basis.

Maple Code: Operator A_ω

```
with(DifferentialGeometry): with(LinearAlgebra): with(Tensor):
```

Coordinates on the 1-jet space:

```
DGsetup( [x1,x2,u,p1,p2], J):
```

Cartan's form and its exterior differential:

```
kappa:=evalDG(du-p1*dx1-p2*dx2):
Omega:=ExteriorDerivative(kappa):
```

Define 2-form ω:

```
omega:=evalDG(2*p1*dp2 &w dx1+ dx1 &w dp1-dx2 &w dp2);
```

Vector fields and 1-forms on Cartan's distribution:

```
VectCartan:=evalDG([D_x1+p1*D_u,D_x2+p2*D_u,D_p1,D_p2]):
CovectCartan:=evalDG([dx1,dx2,dp1,dp2]):
```

Checking duality:

```
m := proc (i, j) options operator, arrow;
    Hook(VectCartan[i],CovectCartan[j])
end proc:
Matrix(4,m):
```

Construct an arbitrary vector field on Cartan's distribution:

```
V:=DGzip([a, b, c,d], VectCartan, "plus"):
```

$$V = a\frac{\partial}{\partial x_1} + b\frac{\partial}{\partial x_2} + (bp_2 + ap_1)\frac{\partial}{\partial u} + c\frac{\partial}{\partial p_1} + d\frac{\partial}{\partial p_2}$$

General form of $A = A_\omega$. Here $a_{i,j}$ are arbitrary functions:

```
A:=evalDG(sum(sum(a[i,j]*VectCartan[i] &t
CovectCartan[j],i=1..4),j=1..4)):
```

Action of A_ω on vector fields:

```
Act:=Z->convert(ContractIndices(evalDG(A &tensor Z),
[[2,3]]), DGvector):
```

Equations w.r.t. $a_{i,j}$:

```
for i from 1 to 4 do
e[i]:=evalDG(Hook(Act(evalDG(VectCartan[i])),Omega)-
Hook(VectCartan[i], omega));
end do:

AEq:=[]:
for i from 1 by 1 to 4 do
AEq:=[op(AEq),op(GetComponents(e[i],CovectCartan))]
end do:
AEq;
```

$$
\begin{array}{llll}
-a_{3,1}, & -a_{4,1}, & -1+a_{1,1}, & 2p_1+a_{2,1}, \\
-a_{3,2}, & -a_{4,2}, & a_{1,2}, & 1+a_{2,2}, \\
1-a_{3,3}, & -a_{4,3}, & a_{1,3}, & a_{2,3}, \\
-2p_1-a_{3,4}, & -1-a_{4,4}, & a_{1,4}, & a_{2,4}
\end{array}
$$

```
sol:=solve(AEq,[a[1,1],a[1,2],a[1,3],a[1,4],
a[2,1],a[2,2],a[2,3],a[2,4],
a[3,1],a[3,2],a[3,3],a[3,4],
a[4,1],a[4,2],a[4,3],a[4,4]]);

assign(sol);

m := proc (i, j) options operator, arrow; a[i,j] end proc;

Am:=Matrix(4,4,m);
```

$$
A_\omega = \begin{Vmatrix} 1 & 0 & 0 & 0 \\ -2p_1 & -1 & 0 & 0 \\ 0 & 0 & 1 & -2p_1 \\ 0 & 0 & 0 & -1 \end{Vmatrix} \tag{36}
$$

```
Determinant(Am);
```

$$1$$

```
Am.Am;
```

$$
\begin{Vmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{Vmatrix}
$$

Characteristic distributions

Effective forms ω and $h\omega$, where h is any nonvanishing function, define the same Monge–Ampère equation. Therefore, for a non-degenerate equation $\mathcal{E}_\omega$ the form ω can be normed in such a way that $|\operatorname{Pf}(\omega)| = 1$. It is sufficient to replace ω by

$$\frac{\omega}{\sqrt{|\operatorname{Pf}(\omega)|}}. \tag{37}$$

By (32), the hyperbolic equations generate a product structure

$$A^2_{\omega,a} = 1$$

and elliptic equations generate a complex structure

$$A^2_{\omega,a} = -1$$

on the Cartan space $C(a)$ [17].

Therefore, a non-degenerate Monge-Ampère equation generates two 2-dimensional (complex — for elliptic case) distributions on J^1M, which are eigenspaces of the operator A_ω.

These distributions $C_+(a)$ and $C_-(a)$ correspond to the eigenvalues 1 and -1 for the hyperbolic equations or to ι and $-\iota$ for the elliptic ones, respectively. Here $\iota = \sqrt{-1}$.

The distributions C_+ and C_- are called *characteristic*.

The characteristic distributions are real for the hyperbolic equations and complex for the elliptic ones. They are complex conjugate for the elliptic equations.

Proposition 3 [17] *1. The characteristic distributions C_+ and C_- are skew-orthogonal with respect to the symplectic structure Ω, i.e. $\Omega(X_+, X_-) = 0$ for $X_\pm \in D(C_\pm)$.*

2. On each of them the 2-form Ω is nondegenerate.

On the other hand, any pair of arbitrary real distributions $C_{1,0}$ and $C_{0,1}$ on J^1M such that

1. $\dim C_{1,0} = \dim C_{0,1} = 2$;
2. $C = C_{1,0} \oplus C_{0,1}$;
3. $C_{1,0}$ and $C_{0,1}$ are skew-orthogonal with respect to the symplectic structure Ω

determines the operator A. Therefore a hyperbolic Monge-Ampère equation can be regarded as such pair $\{C_{1,0}, C_{0,1}\}$ of distributions.

Maple Code: Characteristic distributions

Calculation of eigenvalues and eigenvectors of the operator A_ω:

```
EV,e:=Eigenvectors(Am):
```

Find the vector fields from the Cartan distribution

```
Cp:=[]:Cm:=[]:

for i from 1 to 4 do
if EV[i]=EV[1] then Cp:=[op(Cp),
   (convert((Transpose(e[1..-1,i])),list))]
else
   Cm:=[op(Cm),(convert((Transpose(e[1..-1,i])),list))]
end if
end do:

Vp1:=DGzip(Cp[1], VectCartan, "plus");
Vp2:=DGzip(Cp[2], VectCartan, "plus");
Vm1:=DGzip(Cm[1], VectCartan, "plus");
Vm2:=DGzip(Cm[2], VectCartan, "plus");
```

For example, the characteristic distribution C_+ and C_- of operator (36) are generated by the following vector fields:

$$C_+ = \left\langle p_1 \frac{\partial}{\partial p_1} + \frac{\partial}{\partial p_2}, \quad \frac{\partial}{\partial x_2} + p_2 \frac{\partial}{\partial u} \right\rangle$$

and

$$C_- = \left\langle \frac{\partial}{\partial p_1}, \quad p_1 \frac{\partial}{\partial x_2} + p_1(p_2 - 1)\frac{\partial}{\partial u} - \frac{\partial}{\partial x_1} \right\rangle.$$

Symplectic Monge–Ampère equations

Monge–Ampère equation (17) is called *symplectic* if its coefficients A, B, C, D, E do not depend on v.

In this case, the structures described above (effective differential forms, the differential operator Δ_ω, field of endomorphisms A_ω) can be considered on the 4-dimensional cotangent bundle T^*M instead of the 5-dimensional jet bundle J^1M.

Below we repeat main constructions for the symplectic case.

A smooth function $f \in C^\infty(M)$ defines a section $s_f : M \longrightarrow T^*M$ of the cotangent bundle

$$\pi : T^*M \longrightarrow M$$

by the following formula:

$$s_f : a \longmapsto df_a.$$

Let ω be a differential 2-form on T^*M. Define a differential operator

$$\Delta_\omega : C^\infty(M) \longrightarrow \Omega^2(M), \quad \Delta_\omega(v) := (s_v)^*(\omega).$$

Then equation $\Delta_\omega(v) = 0$ is a symplectic Monge–Ampère equation.

Let Ω be the symplectic structure on T^*M. In canonical coordinates x_1, x_2, p_1, p_2 on T^*M

$$\Omega = dx_1 \wedge dp_1 + dx_2 \wedge dp_2.$$

The differential form ω is said to be *effective* if

$$\omega \wedge \Omega = 0.$$

Pfaffian $\mathrm{Pf}(\omega)$ of the differential 2-form ω is defined by the following equality:

$$\mathrm{Pf}(\omega)\, \Omega \wedge \Omega = \omega \wedge \omega,$$

and formula

$$A_\omega X \rfloor\ \Omega = X \rfloor \omega$$

defines the field of endomorphisms A_ω on T^*M.

The square of operator A_ω is scalar:

$$A_\omega^2 + \mathrm{Pf}(\omega) = 0.$$

Consider now the case when equation is non degenerated, i.e. $\mathrm{Pf}(\omega) \neq 0$ on T^*M. Then the operator A_ω can be normed (see formula (37).

For hyperbolic equations we get almost product structure: $A_\omega^2 = 1$, and for elliptic ones we get almost complex structure: $A_\omega^2 = -1$.

We say that two symplectic equation $\mathcal{E}_\omega$ and $\mathcal{E}_\varpi$ are *symplectically equivalent* if there exist a symplectic transformation Φ such that

$$\Phi^*(\omega) = h\varpi$$

for some function h.

The following theorem gives a criterion of symplectic equivalence of non-degenerated Monge–Ampère equation to linear equations with constant coefficients.

Theorem 3 [18] *Non-degenerated symplectic Monge–Ampère equation $\mathcal{E}_\omega$ is symplectically equivalent to wave equation*

$$v_{xx} - v_{yy} = 0 \tag{38}$$

(in hyperbolic case), or to Laplace equation

$$v_{xx} + v_{yy} = 0$$

(in elliptic case) if and only if the Nijenhuise tensor

$$N_{A_\omega} = 0, \tag{39}$$

where A_ω is the normed operator.

Recall that the Nijenhuise tensor N_A of an operator A is a tensor field of rank (1,2) given by

$$N_A(X,Y) := -A^2[X,Y] + A[AX,Y] + A[X,AY] - [AX,AY]$$

for vector fields X and Y.

Condition (39) can be written in the following equivalent form [19]:

$$d\omega = \frac{1}{2} d\left(\ln|\operatorname{Pf}(\omega)|\right) \wedge \omega.$$

Maple Code: Symplectic equation and Nijenhuis tensor

Below we construct the operator A_ω for nonlinear wave equation

$$v_{xy} = f(x, y, v_x, v_y). \tag{40}$$

Then we calculate the Nijenhuise tensor N_{A_ω} and find conditions under which is this equation symplectically equivalent to the linear wave equation with constant coefficients.

```
with(DifferentialGeometry): with(Tools):
with(PDETools): with(Tensor):with(LinearAlgebra):
DGsetup( [x1,x2,p1,p2], M):

Omega:=evalDG(dx1 &w dp1+dx2 &w dp2):
omega:=evalDG(-2*f(x1,x2,p1,p2)*dx1 &w dx2+
dx1 &w dp1-dx2 &w dp2);

Vect:=evalDG([D_x1,D_x2,D_p1,D_p2]):
Covect:=evalDG([dx1,dx2,dp1,dp2]):

V:=DGzip([a, b, c,d], Vect, "plus"):

A:=evalDG(sum(sum(a[i,j]*(Vect[i] &t
Covect[j]),i=1..4),j=1..4)):

Act:=Z->convert(ContractIndices
(evalDG(A &tensor Z),[[2,3]]),DGvector):

for i from 1 to 4 do
e[i]:=evalDG(Hook(Act(evalDG(Vect[i])),Omega)-
```

```
Hook(Vect[i],omega));
end do:

AEq:=[]:

for i from 1 by 1 to 4 do AEq:=
[op(AEq),op(GetComponents(e[i], Covect))] end do:

sol:=solve(AEq,[a[1,1],a[1,2],a[1,3],a[1,4],
a[2,1],a[2,2],a[2,3],a[2,4],
a[3,1],a[3,2],a[3,3],a[3,4],
a[4,1],a[4,2],a[4,3],a[4,4]]);

assign(sol):
A:=DGsimplify(convert(A, DGtensor)):

N := TensorBrackets(A, A, "Frolicher-Nijenhuis"):

eq:=Tools:-DGinfo(N, "CoefficientSet");

pdsolve(eq);
```

As a result we get:

$$f = F1(x1, x2),$$

where F is an arbitrary function.

So, equation (40) is symplectically equivalent to wave equation (38) if and only if f is a function in x_1 and x_2 only.

Splitting of tangent spaces

Let us return to the space J^1M.

A non-degenerate equation is called *regular* if the derivatives $C_\pm^{(k)}$ $(k = 1, 2, 3)$ of the characteristic distributions are constant rank distributions, too.

Below we consider regular equations only. Then the first derivatives of the characteristic distributions

$$C_\pm^{(1)} := C_\pm + [C_\pm, C_\pm]$$

are three-dimensional. Their intersection

$$l := C_+^{(1)} \cap C_-^{(1)}$$

is a one-dimensional distribution, which is transversal to Cartan distribution.

Therefore, for hyperbolic equations the tangent space $T_a(J^1M)$ splits into the direct sum (see Fig. 4)

$$T_a(J^1M) = C_+(a) \oplus l(a) \oplus C_-(a) \tag{41}$$

at each point $a \in J^1M$ [17].

For elliptic equations we get similar decomposition of the complexification of $T_a(J^1M)$. In this case the distribution l is real, too.



Fig. 4 Splitting of the tangent space $T_a(J^1M)$.

Lecture 5. Tensor invariants of Monge–Ampère equations

Decomposition of de Rham complex

Let us construct the decomposition of the de Rham complex, which is generated by the splitting of tangent spaces.

Decomposition (41) generates a decomposition of the module of exterior s-forms (or its complexification for elliptic equations). Denote the distributions C_+, l, and C_- by P_1, P_2, and P_3, respectively.

Let $D(J^1M)$ be the module of vector fields on J^1M, and let D_j be the module of vector fields tangent to distribution P_j.

Define the following submodules of modules of differential s-forms $\Omega^s(J^1M)$:

$$\Omega_i^s := \{\alpha \in \Omega^s(J^1M) | \ X \rfloor \alpha = 0 \ \forall \ X \in D_j, \ j \neq i\} \quad (i = 1,2,3).$$

Then we get the following decomposition of the module of differential s-forms on J^1M:

$$\Omega^s(J^1M) = \bigoplus_{|\mathbf{k}|=s} \Omega^{\mathbf{k}}, \tag{42}$$

where $\mathbf{k} = (k_1, k_2, k_3)$ is a multi-index, $k_i \in \{0, 1, \ldots, \dim P_i\}$,

$$|\mathbf{k}| = k_1 + k_2 + k_3,$$

and

$$\Omega^{\mathbf{k}} := \left\{ \sum_{j_1+j_2+j_3=|\mathbf{k}|} \alpha_{j_1} \wedge \alpha_{j_2} \wedge \alpha_{j_3}, \text{ where } \alpha_{j_i} \in \Omega_i^{k_i} \right\} \subset \bigotimes_{i=1}^{3} \Omega_i^{k_i}.$$

Three first terms of the decomposition are presented in the diagram (see Fig. 5).

The exterior differential also splits into the direct sum

$$d = \bigoplus_{|\mathbf{t}|=1} d_{\mathbf{t}},$$

where

$$d_{\mathbf{t}} : \Omega^{\mathbf{k}} \to \Omega^{\mathbf{k}+\mathbf{t}}.$$



Fig. 5 Decomposition of de Rham complex.

Theorem 4 [11] *If the multi-index* $\mathbf{t}$ *contains one negative component and this component is* -1 *then the operator* $d_{\mathbf{t}}$ *is a* $C^\infty(J^1M)$*-homomorphism, i.e.,*

$$d_{\mathbf{t}}(f\alpha) = f d_{\mathbf{t}}\alpha \tag{43}$$

for any function f and any differential form $\alpha \in \Omega^{\mathbf{k}}$.

Due to this theorem, we have the seven homomorphisms, and three of them are zeroes. The nontrivial homomorphisms are the following:

$$d_{2,-1,0}, \quad d_{0,-1,2}, \quad d_{-1,1,1} \quad \text{and} \quad d_{1,1,-1}.$$

Tensor invariants

Consider a case

$$\mathbf{t} = \mathbf{1}_j + \mathbf{1}_k - \mathbf{1}_s.$$

Then the differential $d_{\mathbf{t}}$ is a $C^\infty(J^1M)$-homomorphisn. Note that

$$d_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s} : \Omega^{\mathbf{1}_q} \to 0,$$

if $q \neq s$. Then the only non-trivial of $d_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s}$ is the restriction to the module $\Omega^{\mathbf{1}_s}$:

$$d_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s} : \Omega^{\mathbf{1}_s} \to \Omega^{\mathbf{1}_j} \wedge \Omega^{\mathbf{1}_k}.$$

Therefore the homomorphism $d_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s}$ defines a tensor field of the type (2,1). This tensor field we denote by $\tau_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s}$:

$$\tau_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s} \in \Omega^{\mathbf{1}_j} \wedge \Omega^{\mathbf{1}_k} \otimes D_s.$$

A unique non-trivial component of this tensor field is its restriction to $\Omega^{\mathbf{1}_s}$. Note that

$$\tau_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s} : \Omega^{\mathbf{1}_s} \to \Omega^{\mathbf{1}_j} \wedge \Omega^{\mathbf{1}_k}$$

coincides with $d_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s}$.

Tensor fields $\tau_{\mathbf{1}_j+\mathbf{1}_k-\mathbf{1}_s}$ are differential invariants of Monge–Ampère equations. So, we get four tensors of (2,1)-type [11]:

$$\tau_{2,-1,0}, \quad \tau_{0,-1,2}, \quad \tau_{-1,1,1} \quad \text{and} \quad \tau_{1,1,-1}. \tag{44}$$

Maple Code: Tensor invariants

Below we present a program for calculating the tensor $\tau_{-1,1,1}$. The remaining tensors can be found similarly after a small adjustment of the program. In this program, we omit the calculation of the characteristic distributions. They must be calculated in advance (see “Maple Code: Operator A_ω” and “Maple Code: Characteristic distributions”).

```
with(DifferentialGeometry): with(LinearAlgebra):
```

```
with(Tensor):with(Tools): with(PDETools):

DGsetup( [x1,x2,u,p1,p2], J):

kappa:=evalDG(du-p1*dx1-p2*dx2):
Omega:=ExteriorDerivative(kappa):

omega:=evalDG(2*u*dx2 &w dp1+ dx1 &w dp1-
dx2 &w dp2-2*k*p1^2*dx1 &w dx2):
```

Construct the distribution l (transversal to the Cartan distribution). We are looking for l as an intersection of derivatives of the characteristic distributions $C_-^{(1)}$ and $C_+^{(1)}$. This intersection is 1-dimensional and it is generated by the vector field Z which we are looking for.

```
S:=evalDG(a1*Vp1+a2*Vp2+a3*LieBracket(Vp1,Vp2)-
(b1*Vm1+b2*Vm2+b3*LieBracket(Vm1,Vm2))):

sol:=solve(Tools:-DGinfo(S, "CoefficientSet"),
[a1,a2,a3,b1,b2,b3]):

assign(sol):

Z:=evalDG(a1*Vp1+a2*Vp2+a3*LieBracket(Vp1,Vp2)):
```

Basis of the module of vector fields on J^1M and dual basis:

```
BV:=[Vm1,Vm2,Vp1,Vp2,Z]:
BC:=evalDG(DualBasis(BV)):
```

Decomposition of de Rham complex. Bases of $\Omega^1(J^1M)$ and $\Omega^2(J^1M)$:

```
Lambda[1,0,0]:=evalDG([BC[1], BC[2]]);

Lambda[0,1,0]:=evalDG([BC[5]]);

Lambda[0,0,1]:=evalDG([BC[3], BC[4]]);

Lambda[2,0,0]:=evalDG([BC[1] &w BC[2]]); #1

Lambda[1,1,0]:=evalDG([BC[1] &w BC[5], BC[2] &w BC[5]]); #2,3

Lambda[1,0,1]:=evalDG([BC[1] &w BC[3], BC[1] &w BC[4],
BC[2] &w BC[3], BC[2] &w BC[4]]); #4,5,6,7

Lambda[0,1,1]:=evalDG([BC[3] &w BC[5], BC[4] &w BC[5]]); #8,9

Lambda[0,0,2]:=evalDG([BC[3] &w BC[4]]); #10
```

List of elements of the basis of Ω^2:

```
Lambda2:=[op(Lambda[2,0,0]),op(Lambda[1,1,0]),
op(Lambda[1,0,1]), op(Lambda[0,1,1]),op(Lambda[0,0,2])];
```

Construct the tensor $\tau_{-1,1,1}$:

```
unassign('z1','z2','z3','z4','z5','z6','z7','z8','z9','z10');
```

Arbitrary differential 2-form:

```
V:=evalDG(DGzip([z1,z2,z3,z4,z5,z6,z7,z8,z9,z10],
Lambda2, "plus")):
```

Arbitrary 2-form from $\Omega^{1,0,0}$:

```
S:=evalDG(ExteriorDerivative(C1*Lambda[1,0,0][1]+
C2*Lambda[1,0,0][2])-V):

S_coeff:=Tools:-DGinfo(S, "CoefficientSet"):

sol:=solve(S_coeff,{z1,z2,z3,z4,z5,z6,z7,z8,z9,z10});

assign(sol);
```

Projection of a differential 2-form to $\Omega^{0,1,1}$:

```
Pr_011:=evalDG(DGzip([z8,z9],
[Lambda2[8],Lambda2[9]],"plus")):

Pr_011:=convert(Pr_011, DGtensor):

unassign('a','b','c','d'):

Tau:=evalDG(a*Lambda[0,1,1][1] &t BV[1]+
b*Lambda[0,1,1][2] &t BV[1]+
c*Lambda[0,1,1][1] &t BV[2]+
d*Lambda[0,1,1][2] &t BV[2]):

aTau:=ContractIndices(evalDG(Tau &t
(C1*Lambda[1,0,0][1]+C2*Lambda[1,0,0][2])),[[3,4]]):

eq0:=DGsimplify(evalDG(aTau-Pr_011)):
eq:=Tools:-DGinfo(eq0, "CoefficientSet"):
e1:=op(eval(eq,{C1=1,C2=0})):
e2:=op(eval(eq,{C1=0,C2=1})):
```

```
sol:=solve([e1,e2],[a,b,c,d]):
assign(sol):

Tau1:=DGsimplify(Tau):

tau[-1, 1, 1]:=DGsimplify(Tau);
```

Example: Hunter–Saxton equation

Consider the Hunter-Saxton equation

$$v_{tx} = vv_{xx} + \kappa u_x^2, \tag{45}$$

where κ is a constant. This equation is hyperbolic, and it has applications in the theory of liquid crystals [6].

The corresponding effective differential 2-form and the operator A_ω are the following:

$$\omega = 2udq_2 \wedge dp_1 + dq_1 \wedge dp_1 - dq_2 \wedge dp_2 - 2\kappa p_1^2 dq_1 \wedge dq_2$$

and

$$A_\omega = \begin{Vmatrix} 1 & 2u & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & -2\kappa p_1^2 & 1 & 0 \\ 2\kappa p_1^2 & 0 & 2u & -1 \end{Vmatrix}.$$

Let's take the following base in the module of vector fields on J^1M:

$$\begin{aligned} X_1 &= \frac{\partial}{\partial q_1} + p_1 \frac{\partial}{\partial u} + \kappa p_1^2 \frac{\partial}{\partial p_2}, \\ X_2 &= \frac{\partial}{\partial p_1} + u \frac{\partial}{\partial p_2}, \\ Z &= \frac{\partial}{\partial u} + (2\,\kappa - 1)\, p_1 \frac{\partial}{\partial p_2}, \\ Y_1 &= \frac{\partial}{\partial q_2} + \kappa p_1^2 \frac{\partial}{\partial p_1} - u \frac{\partial}{\partial q_1} + (p_2 - u p_1)\, \frac{\partial}{\partial u}, \\ Y_2 &= \frac{\partial}{\partial p_2}. \end{aligned}$$

The dual basis of the module of differential 1-forms is

$$
\begin{aligned}
\alpha_1 &= dq_1 + u dq_2,\\
\alpha_2 &= dp_1 - \kappa p_1^2 dq_2,\\
\theta &= du - p_1 dq_1 - p_2 dq_2,\\
\beta_1 &= dq_2,\\
\beta_2 &= dp_2 + (1-2\kappa)\, p_1 du + (\kappa - 1)\, p_1^2 dq_1 + (2\kappa - 1)\, p_1 p_2 dq_2 - u dp_1.
\end{aligned}
$$

The vector fields X_1, X_2 and Y_1, Y_2 form bases in the modules $D(C_+)$ and $D(C_-)$ respectively. Tensor invariants of equation (45) have the form:

$$
\begin{aligned}
\tau_{-1,1,1} &= -(p_1 dq_1 \wedge dq_2 + dq_2 \wedge du) \otimes \left(\frac{\partial}{\partial q_1} + p_1 \frac{\partial}{\partial u} + \kappa p_1^2 \frac{\partial}{\partial p_2} \right),\\
\tau_{1,1,-1} &= 2(\kappa - 1) \Big(\kappa p_1^3 dq_1 \wedge dq_2 + \kappa p_1^2 dq_2 \wedge du -\\
&\qquad dp_1 \wedge du - p_1 dq_1 \wedge dp_1 - p_2 dq_2 \wedge dp_1 \Big) \otimes \frac{\partial}{\partial p_2},\\
\tau_{2,-1,0} &= \Big(dq_1 \wedge dp_1 - \kappa p_1^2 dq_1 \wedge dq_2 + u dq_2 \wedge dp_1 \Big) \otimes\\
&\qquad \left(\frac{\partial}{\partial u} + (2\kappa - 1)\, p_1 \frac{\partial}{\partial p_2} \right),\\
\tau_{0,-1,2} &= \Big(dq_2 \wedge dp_2 + (1-2\kappa)\, p_1 dq_2 \wedge du + (1-\kappa)\, p_1^2 dq_1 \wedge dq_2 - u dq_2 \wedge dp_1 \Big) \otimes\\
&\qquad \left(\frac{\partial}{\partial u} + (2\kappa - 1)\, p_1 \frac{\partial}{\partial p_2} \right).
\end{aligned}
$$

The Laplace forms

Define bracket $\langle \alpha \otimes X, \beta \otimes Y \rangle$ for decomposable tensors $\alpha \otimes X$ and $\beta \otimes Y$ of types (2,1) as follows [11]:

$$
\langle \alpha \otimes X,\, \beta \otimes Y \rangle = (Y \rfloor \alpha) \wedge (X \rfloor \beta).
$$

For non decomposable tensors the bracket is defined by linearity.

Define two differential 2-forms λ_- and λ_+ from the module $\Omega^{1,0,1}$ as "wedge contractions" of the tensor fields:

$$
\lambda_+ := \left\langle \tau_{0,-1,2},\, \tau_{1,1,-1} \right\rangle, \qquad \lambda_- := \left\langle \tau_{2,-1,0},\, \tau_{-1,1,1} \right\rangle. \tag{46}
$$

Then tensors (46) are called *Laplace forms* of Monge–Ampère equations $\mathcal{E}_\omega$.

Example: Laplace form for linear equations

For linear hyperbolic equation

$$v_{xy} = a(x,y)v_x + b(x,y)v_y + c(x,y)v + g(x,y), \tag{47}$$

the Laplace forms are

$$\lambda_- = k\,dx \wedge dy \quad \text{and} \quad \lambda_+ = -h\,dx \wedge dy, \tag{48}$$

where

$$k = ab + c - b_y \qquad\qquad h = ab + c - a_x \tag{49}$$

are the classical Laplace invariants. This observation justifies our definition.

For linear elliptic equations

$$v_{xx} + v_{yy} = a(x,y)v_x + b(x,y)v_y + c(x,y)v + g(x,y), \tag{50}$$

Laplace forms generalize Cotton invariants [2].

We emphasize that the classical Laplace invariants (49) of equation (50) are not absolute invariants even with respect to transformations

$$\phi : (x,\ y,\ v) \mapsto (X(x), Y(y), A(x,y)v), \qquad A(x,y) \neq 0 \tag{51}$$

in contrast to forms $\lambda_\pm$, which are contact invariants.

Example: Laplace forms for Hunter–Saxton equation

The Laplace forms for the Hunter-Saxton equation (45) are

$$\lambda_- = -dq_2 \wedge dp_1, \quad \lambda_+ = 2\,(1-\kappa)\,dq_2 \wedge dp_1.$$

Contact linearization of the Monge–Ampère equations

It is well known that if the classical Lagrange invariants h and k of a linear hyperbolic equation are zero, then the equation can be reduced to the wave equation (see [**?**], for example).

Similar statement is true for the Monge–Ampère equations [13]:

Theorem 5 *A hyperbolic Monge–Ampère equation is locally contact equivalent to the wave equation*

$$v_{xy} = 0$$

if and only if its Laplace invariants are zero: $\lambda_+ = \lambda_- = 0$.

Corollary 1 *The equation*

$$v_{xy} = f\left(x, y, v, v_x, v_y\right)$$

is locally contact equivalent to the wave equation $v_{xy} = 0$ *if and only if the function* f *has the following form:*

$$f = \varphi_y v_x + \varphi_x v_y + (\varphi_v + \Phi_v) v_x v_y + R,$$

where the function $R = R(x, y, v)$ *satisfies to the following ordinary linear differential equation:*

$$R_v = (\varphi_v + \Phi_v) R + \varphi_{xy} - \varphi_x \varphi_y.$$

Solving this equation we get

$$R = e^{\varphi+\Phi}\left(\int (\varphi_{xy} - \varphi_x\varphi_y) e^{-\varphi-\Phi} dv + g\right),$$

where $\varphi = \varphi(x, y, v)$*,* $\Phi = \Phi(v)$*, and* $g = g(x, y)$ *are arbitrary functions.*

The general problem of linearization of non-degenerated Monge–Ampère equations with respect to contact transformations was solved in [12].

Acknowledgements This work is partially supported by Russian Foundation for Basic Research (project 18-29-10013). The third author was also supported by the Czech Grant Agency, grant nr. GA17-01171S.

References

1. Akhmetzianov, A.V., Kushner, A.G., Lychagin, V.V. Integrability of Buckley–Leverett?s filtration model. In: IFAC Proceedings Volumes (IFAC-PapersOnline). **49**(12), pp. 1251-1254 (2016)
2. Cotton, E.: Sur les invariants différentiels de quelques équations linearies aux dérivées partielles du second ordre. In: Ann. Sci. Ecole Norm. Sup. **17**, pp. 211-244 (1900)
3. Darboux, G.: Leçons sur la théorie générale des surfaces. Vol. I. Paris, Gauthier-Villars, Imprimeur–Libraire, 1887, vi+514 pp.
4. Darboux, G.: Leçons sur la théorie générale des surfaces. Vol. II. Paris, Gauthier-Villars, 1915, 579 pp.
5. Darboux, G.: Leçons sur la théorie générale des surfaces. Vol. III. Paris, Gauthier-Villars at fils, Imprimeur– Libraire, 1894, viii+512 pp.
6. Hunter, J.K., Saxton, R. Dynamics of director fields. In: SIAM J. Appl. Math. **51**(6), pp. 1498-1521 (1991)
7. Kantorovich, L. On the translocation of masses. // C.R. (Doklady) Acad. Sci. URSS (N.S.), 37:199-201 (1942)
8. Kushner, A.G. Classification of mixed type Monge-Ampère equations. In: Pràstaro, A., Rassias, Th.M. (eds) Geometry in Partial Differential Equations, pp. 173–188. World Scientific (1993)

9. Kushner, A.G. Symplectic geometry of mixed type equations. In: Lychagin, V.V. (ed) The Interplay beetween Differential Geometry and Differential Equations. Amer. Math. Soc. Transl. Ser. 2, **167**, pp. 131-142 (1995)
10. Kushner, A.G. Monge–Ampère equations and e-structures. In: Dokl. Akad. Nauk **361**(5), pp. 595-596 (1998) (Russian). English translation in Doklady Mathematics, **58**(1), pp. 103–104 (1998)
11. Kushner, A.G. A contact linearization problem for Monge–Ampère equations and Laplace invariants. In: Acta Appl. Math. **101**(1-3), pp. 177–189 (2008)
12. Kushner, A.G.: Classification of Monge-Ampère equations. In: "Differential Equations: Geometry, Symmetries and Integrability". Proceedings of the Fifth Abel Symposium, Tromso, Norway, June 17–22, 2008 (Editors: B. Kruglikov, V. Lychagin, E. Straume) pp. 223–256.
13. Kushner, A.G.: On contact equivalence of Monge–Ampère equations to linear equations with constant coefficients. Acta Appl. Math. 109(1), 197–210 (2010)
14. Kushner, A.G., Lychagin, V.V., Rubtsov, V.N. Contact geometry and nonlinear differential equations. Encyclopedia of Mathematics and Its Applications **101**, Cambridge University Press, Cambridge, 2007, xxii+496 pp.
15. Lychagin, V.V. Contact geometry and nonlinear second-order partial differential equations. In: Dokl. Akad. Nauk SSSR **238**(5), pp. 273-276 (1978). English translation in Soviet Math. Dokl. **19**(5), pp. 34-38 (1978)
16. Lychagin, V.V. Contact geometry and nonlinear second-order differential equations. In: Uspekhi Mat. Nauk **34**(1 (205)), pp. 137-165 (1979). English translation in Russian Math. Surveys **34**(1), pp. 149-180 (1979)
17. Lychagin, V.V. Lectures on geometry of differential equations. Vol. 1,2. "La Sapienza", Rome, 1993.
18. Lychagin, V.V., Rubtsov, V.N. The theorems of Sophus Lie for the Monge–Ampère equations (Russian). In: Dokl. Akad. Nauk BSSR **27**(5), pp. 396-398 (1983)
19. Lychagin, V.V., Rubtsov, V.N. Local classification of Monge–Ampère differential equations. In: Dokl. Akad. Nauk SSSR **272**(1), pp. 34-38 (1983)
20. Lychagin, V.V., Rubtsov, V.N., Chekalov, I.V. A classification of Monge–Ampère equations. In: Ann. Sci. Ecole Norm. Sup. (4) **26**(3), pp. 281-308 (1993)
21. Tunitskii, D.V. On the contact linearization of Monge-Ampère equations. In: Izv. Ross. Akad. Nauk, Ser. Matem. **60**(2), pp. 195-220 (1996)
22. Krasil'shchik, I.S., Lychagin, V.V., Vinogradov, A.M. Geometry of jet spaces and nonlinear partial differential equations. Advanced Studies in Contemporary Mathematics, **1**, Gordon and Breach Science Publishers, New York, 1986, xx+441 pp.